
Preface

The first edition of An Introduction to Modern Econometrics Using Stata was published
20 years ago, with examples drawn from Stata 9. This second edition was written
using StataNow 19.5 and highlights some of the important features that have been
incorporated into Stata since Stata 9. The growth of Stata’s own capabilities and the
increasing amount of community-contributed commands has made Stata’s role in the
statistical software universe all the more prominent.

In retrospect, the most significant enhancement to Stata’s ease of use was the in-
troduction of factor-variable notation and, along with it, the margins command. More
recently, the introduction of table production commands has addressed one of the key
aspects of researchers’ efficiency: the ability to produce publication-quality results in
various formats. These features, combined with the computational improvements of
embracing 64-bit processors, have played important roles in Stata’s popularity in many
disciplines, including my own fields of applied econometrics and public health.

What follows is drawn from the preface to the first edition of this book. The goals are
largely unchanged, but many aspects of what an applied researcher needs to complete
their tasks have been updated by the evolution of Stata’s capabilities. Much of the
earlier edition has been rewritten to reflect those improvements. I highlight recent
changes, such as the introduction of frames and table production commands. I also
present econometric methods such as the computation of predictive means and average
marginal effects, the correlated random-effects model for panel data, and new methods
for IV estimation. The last chapter, on discrete and limited dependent variables, has
been expanded to include fractional response models and count-data models.

This book is a concise guide for applied researchers in economics, finance, and re-
lated disciplines to learn basic econometrics and work through Stata examples using
typical datasets analyzed in economics. Readers should be familiar with applied statis-
tics at the level of a simple linear regression (ordinary least-squares) model and its
algebraic representation. Knowledge of these concepts would be equivalent to the level
of an undergraduate statistics or econometrics course sequence such as that contained in
Wooldridge (2025). The book also uses some multivariate calculus (partial derivatives)
and linear algebra.

I presume that the reader is familiar with Stata’s windowed interface and with the
basics of data input, data transformation, and descriptive statistics. Readers who are
comfortable interacting with Stata should feel free to skip to chapter 4, where the
discussion of econometrics begins in earnest.



xx Preface

In any research project, much of the effort involves the preparation of the data spec-
ified as part of an econometric model. While this book primarily focuses on applied
econometric practice, we must consider the challenges that many researchers face in
moving from their original data sources to the form needed in an econometric model—
or even the form needed to provide appropriate tabulations and graphs for the project.
Accordingly, chapter 2 focuses on the details of data management and several tools
available in Stata to ensure that the appropriate transformations are accomplished ac-
curately and efficiently. If you are familiar with these aspects of Stata usage, then feel
free to skim this material, perhaps returning to it to refresh your understanding of Stata
usage. Likewise, chapter 3 discusses the organization of economic and financial data and
the Stata commands needed to reorganize data among the several forms of organization.
If you are eager to begin with the econometrics of linear regression, you need only skim
this chapter, noting its content for future reference.

Chapter 4 begins the econometric content of this book and presents the most widely
used tool for econometric analysis: the multiple linear regression model applied to
continuous and categorical variables. The chapter also discusses how to interpret and
present regression estimates and discusses the logic of hypothesis tests and linear and
nonlinear restrictions. The last section of the chapter considers residuals, predicted
values, and marginal effects.

Applying the regression model depends on some assumptions that real datasets often
violate. Chapter 5 discusses how the crucial zero-conditional-mean assumption of the
errors can be violated in the presence of specification errors. The chapter also discusses
statistical and graphical techniques for detecting specification errors. Chapter 6 focuses
on estimation of the variance–covariance matrix, discussing robust, cluster–robust, and
autocorrelation-consistent structures, as well as the generalized linear regression model.
It also explains how to diagnose and correct for heteroskedasticity and serial correlation.
Chapter 7 discusses the usage of indicator or dummy variables in linear regression models
containing both quantitative and qualitative factors, models with interaction effects, and
models of structural change.

Many regression models in applied economics violate the zero-conditional-mean as-
sumption of the errors because they simultaneously determine the response variable and
one or more regressors or because of measurement error in the regressors. No matter the
cause, ordinary least-squares techniques will no longer generate unbiased and consistent
estimates, so you must use instrumental-variables (IV) techniques instead. Chapter 8
presents the IV estimator and its generalized method of moments counterpart along with
tests for determining the need for IV techniques.

Chapter 9 applies models to panel or longitudinal data that have both cross-sectional
and time-series dimensions. Extensions of the regression model allow you to take ad-
vantage of the rich information in panel data, accounting for the heterogeneity in both
panel unit and time dimensions.

Many econometric applications model categorical and limited dependent variables:
a binary outcome—such as a purchase decision, a count, or a constrained response such

as the amount spent—that combines the decision whether to purchase with the decision
of how much to spend, conditional on purchasing. Because linear regression techniques
are generally not appropriate for modeling these outcomes, chapter 10 presents several
limited dependent variable estimators available in Stata.

The appendixes discuss techniques for importing external data into Stata and explain
basic Stata programming. Although you can use Stata without doing any programming,
learning how to program in Stata can help you save a lot of time and effort. You should
also learn to generate reproducible results by using do-files that you can document,
archive, and rerun. Following Stata’s guidelines will make your do-files shorter and
easier to maintain and modify.

Brighton, MA Christopher F. Baum
Oak Square School
September 2026



 

 

 

 

 

 

 



4 Linear regression

This chapter discusses the most widely used tool in applied economics: the linear re-
gression model, which relates a set of continuous variables to a continuous outcome.
The explanatory variables in a regression model often include one or more binary or
indicator variables; see chapter 7. Likewise, many models seek to explain a binary re-
sponse variable as a function of a set of factors, which linear regression does not handle
well. Chapter 10 discusses several forms of that model, including those in which the
response variable is limited but not binary.

4.1 Introduction
This chapter discusses multiple regression in the context of a prototype economic re-
search project. To carry out such a research project, we must

1. define a research framework—or economic model—that lets us specify the ques-
tions of interest and defines how we will interpret the empirical results;

2. find a dataset containing empirical counterparts to the quantities specified in the
economic model;

3. use exploratory data analysis to familiarize ourselves with the data and identify
outliers, extreme values, and the like;

4. fit the model and use specification analysis to determine the adequacy of the
explanatory factors and their functional form;

5. conduct statistical inference (given satisfactory findings from specification analy-
sis) on the research questions posed by the model; and

6. analyze the findings from hypothesis testing and the success of the model in terms
of predictions and marginal effects. Based on these findings, we may have to
return to one of the earlier stages to reevaluate the dataset and its specification
and functional form.

Section 4.2 reviews the basic regression analysis theory on which regression point
and interval estimates are based. Section 4.3 introduces a prototype economic research
project studying the determinants of wage determination and discusses the various com-
ponents of Stata’s results from fitting a regression model of wages. Section 4.4 discusses
how to transform Stata’s estimation results into publication-quality tables. Section 4.5
discusses hypothesis testing and estimation subject to constraints on the parameters.

109

Section 4.6 deals with computing residuals and predicted values. The last section dis-
cusses computing marginal effects. In the following chapters, we take up violations of
the assumptions on which regression estimates are based.

4.2 Computing linear regression estimates
The linear regression model is the most widely used econometric model and the baseline
against which all others are compared. It specifies the conditional mean of a response
variable y as a linear function of k independent variables

E (y | x1, x2, . . . , xk) = β1x1 + β2x2 + · · ·+ βkxk

Given values for the βs, which are fixed parameters, the regression function predicts the
average value of y in the population for different values of x1, x2, . . . , xk.

Suppose that the mean value of an individual’s wage, conditional on their years of
work experience, is given by

E (wage | ttl_exp) = β1 + β2 ttl_exp

where ttl_exp is years of their work experience. This relationship reflects the hypothesis
that wages will generally increase with human capital formation, captured by their
experience in the labor market.

Figure 4.1. Conditional mean of wage

Figure 4.1 shows hourly wages recorded for 2,246 individuals, along with the linear
fit of wages to years of work experience. The regression function’s estimate of the
conditional mean of wage for each value of ttl_exp is shown by the appropriate point

on the line. As theory predicts, wages rise with human capital formation. Of course,
this relationship between the wage and years of experience must be considered ceteris
paribus: all other factors that might affect the individual’s wage are held constant when
we evaluate the effect of additional years of work experience on his or her wage.

When we are working with economic data, we do not know the population values of
β1, β2, . . . , βk. We work with a sample of N observations of data from that population.
Using the information in this sample, we must

1. obtain consistent estimates of the coefficients (β1, β2, . . . , βk),
2. determine how much our coefficient estimates would change if we were given an-

other sample from the same population,
3. decide whether there is enough evidence to rule out some values for some of the

coefficients (β1, β2, . . . , βk), and
4. use our estimated (β1, β2, . . . , βk) to interpret the model.

To obtain estimates of the coefficients, some assumptions must be made about the
process that generated the data. I discuss those assumptions below and describe what I
mean by good estimates. Before performing steps 2–4, I check whether the data support
these assumptions by using a process known as specification analysis.

If we have a cross-sectional sample from the population, the linear regression function
for each observation in the sample has the form

yi = β1 + β2xi,2 + β3xi,3 + · · ·+ βkxi,k + ui

for each observation in the sample i = 1, 2, . . . , N . The u process is a stochastic distur-
bance representing the net effect of all other unobservable factors that might influence
y. The variance of its distribution, σ2

u, is an unknown population parameter to be
estimated along with the β parameters. We assume that N > k; to conduct statis-
tical inference, there must be more observations in the sample than parameters to be
estimated. In practice, N should be much larger than k.

We can write the linear regression function in matrix form as

y = Xβ + u (4.1)

where X is an (N × k) matrix of sample values.1

This population regression function specifies that a set of k regressors in X and the
stochastic disturbance u are the determinants of the response variable (or regressand)
y. We usually assume that the regression function contains a constant term, so x1 is
understood to equal 1 for each observation.

1. Some textbooks use k in this context to refer to the number of slope parameters rather than the
number of columns of X. That will explain the deviations in the formulas given below; where I
write k, some authors write (k + 1).

The key assumption in the linear regression model involves the relationship in the
population between the regressors x and u.2 We may rewrite (4.1) as

u = y − xβ

We assume that
E (u | x) = 0 (4.2)

that is, that the u process has a zero conditional mean. This assumption is that the un-
observed factors involved in the regression function are not related systematically to the
observed factors. This approach to linear regression lets us consider both nonstochastic
and stochastic regressors in X without distinction as long as they satisfy assumption
(4.2).3

We also assume conditional homoskedasticity of the error process:

E
(
u2 | x

)
= σ2

u (4.3)

This assumption states that the error variance, conditional on the regressors, is constant.
Given the zero-conditional-mean assumption, this also implies that the unconditional
expectation of u2 is σ2

u. This assumption is testable, as we will see in our description
of specification analysis. The assumption of conditional homoskedasticity also implies
that

Var (y | x) = σ2
u

4.2.1 Regression as a method of moments estimator

We can use the zero-conditional-mean assumption shown in (4.2) to define a method
of moments estimator of the regression function. Method of moments estimators are
defined by moment conditions that are assumed to hold for the population moments.
When we replace the unobservable population moments by their sample counterparts,
we derive feasible estimators of the model’s parameters. The zero-conditional-mean
assumption gives rise to a set of k moment conditions, one for each x. In particular, the
zero-conditional-mean assumption implies that each regressor is uncorrelated with u.4

E(x′u) = 0
E{x′(y − xβ)} = 0 (4.4)

2. x is a vector of random variables, and u is a scalar random variable. In (4.1), X is a matrix of
realizations of the random vector x, and u and y are vectors of realizations of the scalar random
variables u and y.

3. Chapter 8 discusses how to use the instrumental-variables estimator when the zero-conditional-mean
assumption is violated.

4. The assumption of a zero conditional mean is stronger than that of a zero covariance because
covariance considers only linear relationships between the random variables.

Substituting calculated moments from our sample into the expression and replacing the
unknown coefficients β with estimated values β̂ in (4.4) yields the ordinary least-squares
(OLS) estimator

X′y − X′Xβ̂ = 0

β̂ = (X′X)−1X′y (4.5)

We can use β̂ to calculate the regression residuals:

û = y − Xβ̂

Given the solution for the vector β̂, the additional parameter of the regression prob-
lem σ2

u—the population variance of the stochastic disturbance—can be estimated as a
function of the regression residuals ûi,

s2 =

∑N
i=1 û

2
i

N − k
=

û′û
N − k

(4.6)

where (N−k) is the residual degrees of freedom of the regression problem. The positive
square root of s2 is often termed the standard error of regression, or root mean squared
error. Stata uses the latter term and displays s as Root MSE.

The method of moments is not the only approach for deriving the linear regression
estimator of (4.5), which is the well-known formula from which the OLS estimator is
derived.5

4.2.2 The sampling distribution of regression estimates

The OLS estimator β̂ is a vector of random variables because it is a function of the
random variable y, which in turn is a function of the stochastic disturbance u. The OLS
estimator takes on different values for each sample of N observations drawn from the
population. Because we often have only one sample to work with, we may be unsure of
the usefulness of the estimates from that sample. The estimates are the realizations of
the random vector β̂ from the sampling distribution of the OLS estimator. To evaluate
the precision of a given vector of estimates β̂, we use the sampling distribution of the
regression estimator.

To learn more about the sampling distribution of the OLS estimator, we must make
further assumptions about the distribution of the stochastic disturbance ui. In clas-
sical statistics, the ui were assumed to be independent draws from the same normal
distribution. The modern approach to econometrics drops the normality assumption
and simply assumes that the ui satisfy the assumptions of a zero conditional mean and
conditional homoskedasticity.

5. The treatment here is similar to that of Wooldridge (2025). See Stock and Watson (2019) for a
derivation based on minimizing the squared-prediction errors.



114 Chapter 4 Linear regression

The exact finite-sample distribution of the OLS estimator can be derived by using
the normality assumption. In contrast, under these weaker assumptions, we must use
large-sample theory to derive the sampling distribution of the OLS estimator. Basically,
large-sample theory supposes that the sample size N becomes infinitely large. Because
no real sample is infinitely large, these methods approximate the sampling distribution
of the OLS estimator only in finite samples. With a few hundred observations or more,
the large-sample approximation works well, so these methods work well with applied
economic datasets.

Although large-sample theory is more abstract than finite-sample methods, it im-
poses weaker assumptions on the data-generating process. We will use large-sample
theory to define “good” estimators and to evaluate the precision of the estimates pro-
duced from a given sample.

In large samples, consistency means that as N goes to ∞, the estimates will converge
to their respective population parameters. Roughly speaking, if the probability that the
estimator produces estimates arbitrarily close to the population values goes to 1 as the
sample size increases to infinity, the estimator is said to be consistent.

The sampling distribution of an estimator describes the set of estimates produced
when that estimator is applied to repeated samples from the underlying population.
You can use the sampling distribution of an estimator to evaluate the precision of a
given set of estimates and to statistically test whether the population parameters take
on certain values.

Large-sample theory shows that the sampling distribution of the OLS estimator is
approximately normal.6 Specifically, when the ui are conditionally homoskedastic with
finite variance σ2

u, the OLS estimator β̂ has a large-sample normal distribution with
mean β and variance σ2

uQ−1, where Q−1 is the variance–covariance matrix of X in the
population. The variance–covariance matrix of the estimator, σ2

uQ−1, is also referred to
as a VCE. Because it is unknown, we need a consistent estimator of the VCE. Although
neither σ2

u nor Q−1 is actually known, we can use consistent estimators for them to
construct a consistent estimator of σ2

uQ−1. Given that s2 consistently estimates σ2
u and

(1/N)(X′X) consistently estimates Q, s2(X′X)−1 is a VCE of the OLS estimator.7

4.2.3 Efficiency of the regression estimator

Under the assumption of conditionally homoskedastic errors, the Gauss–Markov theo-
rem holds. Among linear, unbiased estimators of β, the OLS estimator has the smallest

6. More precisely, the distribution of the OLS estimator converges to a normal distribution. In
this book, I will simply refer to the “approximate” or “large-sample” normal distribution. See
Wooldridge (2025, chap. 5) for an introduction to large-sample theory.

7. At first glance, you might think that the expression for the VCE should be multiplied by 1/N ,
but this assumption is incorrect. Because the OLS estimator is consistent, it is converging to the
constant vector of population parameters at the rate 1/

√
N , implying that the variance of the OLS

estimator is going to 0 as the sample size gets larger. Large-sample theory compensates for this
effect in how it standardizes the estimator. The loss of the 1/N term in the estimator of the VCE
is a product of this standardization.

sampling variance, or the greatest precision.8 In that sense, it is best, so OLS is “BLUE”
(the best linear unbiased estimator) for the parameters of the regression model. If we
consider only unbiased estimators that are linear in the parameters, we cannot find a
more efficient estimator. The property of efficiency refers to the precision of the estima-
tor. If estimator A has a smaller sampling variance than estimator B, then estimator
A is said to be relatively efficient. The Gauss–Markov theorem states that OLS is rela-
tively efficient versus all other linear, unbiased estimators of the model. However, this
statement rests upon the hypotheses of an appropriately specified model and an error
process with a zero conditional mean, as specified in (4.2). Note that although unbi-
asedness does not rely upon the assumption of conditional homoskedasticity in (4.3), it
is required to achieve the BLUE result of the Gauss–Markov theorem.

4.2.4 Numerical identification of the regression estimates

As in (4.5) above, the solution to the regression problem involves a set of k moment
conditions, or equations to be jointly solved for the k parameter estimates β̂1, β̂2, . . . , β̂k.
When will these k parameter estimates be uniquely determined, or numerically iden-
tified? We must have more sample observations than parameters to be estimated; in
other words, N must be greater than k. That condition is not sufficient, though. For
the simple “two-variable” regression model yi = β1+β2xi,2+ui, Var(x2) must be greater
than 0. If there is no variation in x2, the data do not provide sufficient information to
determine estimates of β1 and β2.

In multiple regression with many regressors, XN×k must be a matrix of full column
rank k, which implies two things. First, only one column of X can take on a constant
value, so each other regressor must have a positive sample variance. Second, there are no
exact linear dependencies among the columns of the matrix X. The assumption that X is
of full column rank is often stated as “(X′X) is of full rank” or “(X′X) is nonsingular (or
invertible)”. If the matrix of regressors X contains k linearly independent columns, the
cross-product matrix (X′X) will have rank k, its inverse will exist, and the parameters
β1, . . . , βk in (4.5) will be numerically identified.9 If numerical identification fails, the
sample does not contain enough information for us to use the regression estimator on
the model as it is specified. That regression function may be valid as a description of the
data-generating process, but the particular sample may lack the necessary information
to generate a regressor matrix of full column rank. Then we must either respecify
the model or acquire another sample that contains the information needed to uniquely
determine the regression estimates.

8. For a formal presentation of the Gauss–Markov theorem, see any econometrics text, for example,
Wooldridge (2025, 95–96). The OLS estimator is said to be unbiased because E(β̂) = β.

9. When computing infinite precision, we must be concerned with numerical singularity and a computer
program’s ability to reliably invert a matrix regardless of whether it is analytically invertible. As we
discuss in section 4.3.6, computationally near-linear dependencies among the columns of X should
be avoided.

4.3 Interpreting regression estimates
This section illustrates using regression by an example from a prototype research project
and discusses how Stata presents regression estimates. Then it discusses how to recover
the information displayed in Stata’s estimation results for further computations within
your program and how to combine this information with other estimates to present
them in a table. The last subsection considers problems of numerical identification, or
collinearity, that could appear when you are estimating the regression equation.

4.3.1 Research project: A study of wage determination

As an illustration, I present regression estimates from a model fit to 2,246 individuals’
wage data, in which the response variable is the wage in US dollars per hour in 1988.
These data are extracted from the National Longitudinal Survey of Young Women, aged
14–24 in 1968.

nlsw88.dta contains individuals’ attributes that we expect would influence the wage
they earn. Our research question relates to the effect of each of these attributes on wages.
We explore the variables available in these data with describe (see [D] describe). The
notes (see [D] notes) command provides further information:

. webuse nlsw88
(NLSW, 1988 extract)
. describe
Contains data from https://www.stata-press.com/data/r19/nlsw88.dta
Observations: 2,246 NLSW, 1988 extract

Variables: 17 1 May 2220 22:52
(_dta has notes)

Variable Storage Display Value
name type format label Variable label

idcode int %8.0g NLS ID
age byte %8.0g Age in current year
race byte %8.0g racelbl Race
married byte %8.0g marlbl Married
never_married byte %16.0g nev_mar Never married
grade byte %8.0g Current grade completed
collgrad byte %16.0g gradlbl College graduate
south byte %9.0g southlbl Lives in the south
smsa byte %9.0g smsalbl Lives in SMSA
c_city byte %16.0g ccitylbl Lives in a central city
industry byte %23.0g indlbl Industry
occupation byte %22.0g occlbl Occupation
union byte %8.0g unionlbl Union worker
wage float %9.0g Hourly wage
hours byte %8.0g Usual hours worked
ttl_exp float %9.0g Total work experience (years)
tenure float %9.0g Job tenure (years)

Sorted by: idcode

. notes
_dta:

1. 1988 data, extracted from National Longitudinal Survey of Young Women who
were ages 14-24 in 1968 (NLSW).

2. This dataset is the result of extraction and processing by various people
at various times.

3. For more information on the NLS, see http://www.bls.gov/nls/.

We note that most of the variables are of int or byte datatype, with several ex-
pressed as float values, containing a decimal point. Several of the integer variables
also have value labels, identifying the coding of variables such as race.

We present the descriptive statistics with summarize:

. summarize
Variable Obs Mean Std. dev. Min Max

idcode 2,246 2612.654 1480.864 1 5159
age 2,246 39.15316 3.060002 34 46
race 2,246 1.282725 .4754413 1 3

married 2,246 .6420303 .4795099 0 1
never_marr~d 2,246 .1041852 .3055687 0 1

grade 2,244 13.09893 2.521246 0 18
collgrad 2,246 .2368655 .4252538 0 1

south 2,246 .4194123 .4935728 0 1
smsa 2,246 .7039181 .4566292 0 1

c_city 2,246 .2916296 .4546139 0 1

industry 2,232 8.189516 3.010875 1 12
occupation 2,237 4.642825 3.408897 1 13

union 1,878 .2454739 .4304825 0 1
wage 2,246 7.766949 5.755523 1.004952 40.74659
hours 2,242 37.21811 10.50914 1 80

ttl_exp 2,246 12.53498 4.610208 .1153846 28.88461
tenure 2,231 5.97785 5.510331 0 25.91667

The statistics reveal that many of the integer variables are indicators, or binary vari-
ables, while others take on multiple values. The union variable has 368 missing values.
Because most statistical commands use casewise deletion (removing any observation for
which any variable is coded as missing), the number of observations used in a model
including union will be limited. All the variables take on sensible values, with the pos-
sible exception of grade, which should not be 0. To investigate, we use tabulate to
examine its distribution:

. tabulate grade
Current
grade

completed Freq. Percent Cum.

0 2 0.09 0.09
4 3 0.13 0.22
5 1 0.04 0.27
6 14 0.62 0.89
7 19 0.85 1.74
8 33 1.47 3.21
9 55 2.45 5.66
10 84 3.74 9.40
11 123 5.48 14.88
12 943 42.02 56.91
13 176 7.84 64.75
14 187 8.33 73.08
15 92 4.10 77.18
16 252 11.23 88.41
17 106 4.72 93.14
18 154 6.86 100.00

Total 2,244 100.00

We conjecture that it is implausible that respondents received less than a primary
education. Therefore, we change values that are less than 6 to missing:

. replace grade = . if grade<6
(6 real changes made, 6 to missing)

We can also examine more detailed summary statistics for our outcome variable,
wage:

. summarize wage, detail
Hourly wage

Percentiles Smallest
1% 1.930993 1.004952
5% 2.801002 1.032247
10% 3.220612 1.151368 Obs 2,246
25% 4.259257 1.344605 Sum of wgt. 2,246
50% 6.27227 Mean 7.766949

Largest Std. dev. 5.755523
75% 9.597424 40.19808
90% 12.77777 40.19808 Variance 33.12604
95% 16.52979 40.19808 Skewness 3.096199
99% 38.70926 40.74659 Kurtosis 15.85446

The variable is heavily skewed with a long right tail, as expected. This can be viewed
using the kdensity (see [R] kdensity) command, as shown in figure 4.2:

. kdensity wage

Figure 4.2. Density estimate of wage

Because analysis of wages often makes use of a log-linear model, we transform the
wage with logarithms and reevaluate the density plot in figure 4.3:

. generate lwage = log(wage)

. label variable lwage "log(wage)"

. kdensity lwage, normal legend(position(6) rows(1))

Figure 4.3. Density estimate of log(wage)

Using the normal option of kdensity, we see that although wage is highly skewed,
the distribution of lwage is much closer to that of a normally distributed random vari-
able. Although we can estimate a log-linear regression, the generalized linear model
(see [R] glm) is an alternative estimator that may be preferable. We will discuss this
model in section 5.3.

We may also want to explore the frequency distributions of some of the integer vari-
ables, such as race and age, using the tabulate twoway and table (see [R] tabulate
twoway and [R] table) commands:

. tabulate age race
Age in
current Race

year White Black Other Total

34 37 15 1 53
35 183 74 3 260
36 184 71 2 257
37 146 75 4 225
38 158 59 2 219
39 165 67 2 234
40 167 41 0 208
41 162 56 4 222
42 121 37 2 160
43 129 33 3 165
44 117 43 3 163
45 66 12 0 78
46 2 0 0 2

Total 1,637 583 26 2,246
. table (age) (race), stat(percent,across(race)) nototal

Race
White Black Other

Age in current year
34 69.81 28.30 1.89
35 70.38 28.46 1.15
36 71.60 27.63 0.78
37 64.89 33.33 1.78
38 72.15 26.94 0.91
39 70.51 28.63 0.85
40 80.29 19.71
41 72.97 25.23 1.80
42 75.62 23.12 1.25
43 78.18 20.00 1.82
44 71.78 26.38 1.84
45 84.62 15.38
46 100.00

We see that Black respondents have a smaller share of the sample for higher age
groups.

The table command can also be used to produce three-way tables:

. table (married collgrad) (race), nototal

Race
White Black Other

Married
Single

College graduate
Not college grad 355 256 5
College grad 132 53 3

Married
College graduate
Not college grad 862 224 12
College grad 288 50 6

We can also generate tables of summary statistics with table:

. table (result) (race), stat(mean wage) stat(sd wage) stat(p50 wage)
> nformat(%5.2f)

Race
White Black Other Total

Mean 8.08 6.84 8.55 7.77
Standard deviation 5.96 5.08 5.21 5.76
50th percentile 6.55 5.43 7.56 6.27

We see that White respondents have a considerably higher mean wage than Black
respondents. This also holds for the median (p50) wage.

Now that we have familiarized ourselves with the data, we fit a multiple regression
of wage on three continuous variables: ttl_exp, tenure, and grade. We use the i.
factor-variable prefix on a list of integer variables to be included: race, south, smsa,
and union. The regress (see [R] regress) command, like other Stata estimation com-
mands, requires us to specify the response variable followed by a list of the explanatory
variables:10

. regress wage ttl_exp tenure grade i.(race south smsa union)
Source SS df MS Number of obs = 1,861

F(8, 1852) = 112.00
Model 10538.705 8 1317.33812 Prob > F = 0.0000

Residual 21783.4718 1,852 11.7621338 R-squared = 0.3261
Adj R-squared = 0.3231

Total 32322.1768 1,860 17.3775144 Root MSE = 3.4296

wage Coefficient Std. err. t P>|t| [95% conf. interval]

ttl_exp .232707 .0215614 10.79 0.000 .1904198 .2749942
tenure .0522526 .0175946 2.97 0.003 .0177453 .08676
grade .5527474 .0334101 16.54 0.000 .487222 .6182728

race
Black -.6309841 .1929753 -3.27 0.001 -1.009456 -.2525121
Other .5670115 .7232603 0.78 0.433 -.8514796 1.985503

south
South -.6525186 .1726751 -3.78 0.000 -.9911769 -.3138603

smsa
SMSA 1.51067 .1805822 8.37 0.000 1.156504 1.864836

union
Union .7640577 .1903099 4.01 0.000 .3908131 1.137302
_cons -3.850026 .4807968 -8.01 0.000 -4.792987 -2.907066

The header of the regression output describes the overall model estimates, whereas
the table presents the point estimates, their precision, and their interval estimates.

4.3.2 The ANOVA table: ANOVA F and R2

The regression output for this model includes the analysis of variance (ANOVA) table in
the upper left, where the two sources of variation are displayed as Model and Residual.
The SS column shows the sums of squares, with the Residual SS corresponding to
û′û and the Total SS to ỹ′ ỹ in (4.7) below. The df column reports the degrees of
freedom associated with each sum of squares. The degrees of freedom for the total SS
is (N − 1) because the total SS have been computed by using one sample statistic, y.
The degrees of freedom for the model is (k − 1), equal to the number of slopes (or
explanatory variables), or one fewer than the number of estimated coefficients due to
the constant term. Model SS refers to the ability of the regressors to jointly explain a

10. Without any explanatory variables, regress estimates the sample mean of the response variable.

fraction of the variation of y about its mean (Total SS). The residual degrees of freedom
is (N − k), indicating that (N − k) residuals may be freely determined and still satisfy
the constraint from the first normal equation of least squares that the regression surface
passes through the multivariate point of means (y, x2, . . . , xk):

y = β̂1 + β̂2x2 + β̂3x3 + · · ·+ β̂kxk

In the presence of the constant term β̂1, the first normal equation implies that
û = y − Σixiβ̂i must be identically 0.11 This is not an assumption but rather an
algebraic implication of the least-squares technique, which guarantees that the sum and
mean of least-squares residuals will be close to 0.12

The MS column reports the mean squares due to regression and error, or the SS
divided by the df. The ratio of the Model MS to Residual MS is reported as the ANOVA
F statistic, with numerator and denominator degrees of freedom equal to the respective
df values. This ANOVA F statistic is a test of the null hypothesis13 that the slope
coefficients in the model are jointly 0; that is, the null model of yi = µ+ui is as successful
in describing y as is the regression alternative. Prob > F is the tail probability or p-
value of the F statistic. Here we can reject the null hypothesis at any conventional
level of significance. Also, the Root MSE for the regression of 3.43, which is in the units
of the response variable y, can be compared with the mean of that variable, 7.60. We
can use the estat summarize postestimation command to evaluate the values of the
observations included in the model because the inclusion of union has reduced the size
of the estimation sample to 1,861 observations.

11. Recall that the first column of X = ι, an N -element unit vector.
12. Because computers use finite arithmetic, the sum will differ from 0. A well-written computer

program should result in a difference similar to machine precision. For this regression, Stata reports
a mean residual of −1.4× 10−15, comparable with the epsdouble() value of 2.2× 10−16, which is
the smallest number distinguishable by Stata.

13. I discuss hypothesis testing in detail in section 4.5.

. estat summarize
Estimation sample regress Number of obs = 1,861

Variable Mean Std. dev. Min Max

wage 7.600456 4.168635 1.344605 39.23074
ttl_exp 12.85004 4.596992 .1153846 28.88461
tenure 6.571557 5.638432 0 25.91667
grade 13.19882 2.494557 6 18

race
Black .2649113 .4414046 0 1
Other .0123589 .1105114 0 1

south
South .424503 .4944002 0 1

smsa
SMSA .7082214 .4547031 0 1

union
Union .2466416 .4311721 0 1

The upper-right section of the regress output contains several goodness-of-fit statis-
tics, which measure the degree to which a fitted model can explain the variation of the
response variable y. All else equal, we should prefer a simpler model with a better fit
to the data. The mechanics of regression imply that a model with many regressors can
explain y arbitrarily well. Given the least-squares residuals, the most common mea-
sure of goodness of fit, regression R2, may be calculated (given a constant term in the
regression function) as

R2 = 1− û′û
ỹ′ ỹ (4.7)

where ỹ = y − y: the regressand with its sample mean removed. This calculation
emphasizes that the object of regression is not to explain y′y, which is the raw sum of
squares of the response variable y. This would merely explain why E(y) 6= 0—not an
interesting question. Rather, we want to explain the variation in the response variable
around its sample mean.

With a constant term in the model, the least-squares approach seeks to explain
the largest possible fraction of the sample variation of y about its mean and not the
associated variance. The null model with which (4.1) is contrasted is y = µ+ ui, where
µ is the population mean of y. When we fit a regression, we want to determine whether
the information in the regressors x is useful. Is the conditional expectation E(y|x) more
informative than the unconditional expectation E(y) = µ? The null model above has an
R2 = 0, whereas virtually any set of regressors will explain some fraction of the variation
of y around y, the sample estimate of µ. R2 is that fraction in the unit interval, the
proportion of the variation in y about y explained by x.



 

 

 

 

 

 

 



